

# **LIS Self Teaching Package 2018**

**SAS version**

**Part II: Gender, employment, and  
wages**





## **Part II**

### **Gender, employment, and wages**

#### **Overall Plan and Structure of the Exercise**

The exercises in Part I demonstrated the use of household income data along with useful programming techniques for working with the LIS data. Part II emphasises the use of person-level data, including wages, demographics, and labour market information. Whereas Part I consisted entirely of calculating descriptive statistics, Part II introduces regression modelling in the final exercises.

The program that was written in the first set of exercises is now completed and can be set aside. Starting with the next exercise, you will begin the process of building up an entirely new program for Part II. Many of the techniques shown in the previous part will be useful again. In addition, users will learn how to combine LIS datasets by merging household and person files, and by concatenating multiple country-year datasets into a single file.

The general approach of the exercises is the same as in Part I. After beginning a new program in the initial exercise, each subsequent exercise will add new code to the existing program. Within each exercise, results will be produced that help to illuminate the central research themes of this section.

#### **Research Questions**

The analysis of poverty and inequality using household income, which was covered in Part 1, has always been central to research using LIS data. Over the years, however, there has been an increasing volume of work that examines individual outcomes in the labor market. The richness of the labor market data available in LIS has increased over time, and today it is possible to address many types of questions about wages and employment.

Labor market outcomes for women are one especially popular area of research. Women's rate and intensity of work shows much wider cross-country

variation than men's. At the same time, on average women consistently earn lower wages.

One reflection of this growing body of research is the recent creation of the LIS *Employment Key Figures by Gender* (<http://www.lisdatacenter.org/lis-ekf-webapp/app/search-ekf-tables>), which provide information about demographics, employment, earnings, and inequality by both country and gender. In the following exercises, we will investigate several of the topics covered by the key figures, although our results will differ somewhat due to sample selection and variable coding made for the exercises.

For the exercises, we will examine three countries, using data from LIS Wave VI: the United States, Germany, and Greece. As we will see, labor market outcomes for women show distinctly different patterns in each of these countries. Looking at persons of prime working age (which we will define as ages 25-54), our central questions will be:

- How does the percentage of prime-aged women employed in paid work vary across these three countries?
- Among those who are employed, how does the rate of *part-time* employment among women vary across the countries?
- How does employment vary by partnership and family status?
- How do wage differentials between men and women vary between across countries, across levels of educational attainment, and between immigrants and non-immigrants?

In the exercises, we will begin by producing tabulations of employment and wages for various population subgroups. In the concluding exercises, we will use linear regression to study multiple determinants of wages simultaneously, in order to better understand how family structure, education, and immigrant status are related to wages for men and for women.

## **Contents**

### **1. Merging person and household data, selecting a sample**

- Selecting a sample of prime-age workers
- Homeownership rates by country

### **2. Stacking data, employment rates by gender**

- Employment rates by country and gender
- Part-time employment rates by country and gender

### **3. Family structure and employment**

- Family/partnership status and employment rates

### **4. Dependent employment and hourly wages**

- Non-dependent employment as a proportion of employment, by country
- Gender wage gaps among dependent employees, by country

### **5. Hourly wages, education, and country-specific variables**

- Harmonised vs. country-specific codings
- Gender wage gaps by country and educational attainment

### **6. Immigration and wages, understanding harmonisation**

- Understanding harmonization choices
- Gender wage gaps by country and immigrant status

### **7. Wage regressions**

- Regressing log wages on demographics, separately by gender and country

### **8. Pooled regressions and normalised weights**

- PPP adjustments for income
- Pooled regression with multiple countries, using normalised weights

# 1. Merging person and household data, selecting a sample

## Goal

While the exercises in Part I used only data at the household level, Part II uses data from both the household and person level files. In this exercise, we will combine the person and household files in order to create a single dataset for each country in which household data is appended to each person record.

This exercise also selects the universe of persons that we will be studying in the subsequent exercises. Since we are interested in labor market outcomes, we will restrict our attention to people of *prime age*: those who are likely to be old enough to have completed schooling and young enough to not yet be retired. In these exercises, we will define prime age persons as those between 25 and 54 years old. This is a commonly used range in statistics from the United States government and other sources, but other definitions are also possible.

Some of the variables we will be using are not always available for household members other than the head and spouse. For that reason, we will further restrict our universe to heads and partners only.

All of the variables that will be needed in the subsequent exercises are introduced here. However, for now we will only analyze one: we will summarize home ownership, which is a household-level variable, in order to measure the rate of homeownership in each of the three countries under analysis. Homeownership will be included in our later multivariate analysis, because it serves as a rough proxy for wealth, which we otherwise have no information about.

## Activity

Write a program to loop through three datasets: United States 2004 (**us04**), Germany 2004 (**de04**), and Greece 2004 (**gr04**). In each country, merge the person file to the household file and keep the following variables:

- Household: Unique household identifier (**hid**) and owned/rented housing indicator (**own**)
- Person: Unique household identifier (**hid**), dataset name (**dname**), normalized person weight (**pwgt**), inflated person weight (**ppopwgt**), relationship to the household head (**relation**), partnership status

(**partner**), living with own children (**children**), age (**age**), sex (**sex**), immigrant indicator (**immigr**), 3-category recoded educational attainment (**educ**), country-specific educational attainment (**educ\_c**), indicator for employment (**emp**), status in employment (**status1**), indicator for part-time employment (**ptime**), and gross hourly wage in the first job (**gross1**).

Keep only those cases that are in the prime age range (between 25 and 54), and which are defined as either household heads or spouses in the variable **relation**.

Create an indicator variable equal to 1 if a person owns their house (with or without a mortgage), and 0 otherwise. Summarize this new variable to find the homeownership rate among the prime-aged persons for each country, and complete the following table:

Dataset	Homeownership %
US04	
DE04	
GR04	

**Question:** In which country do the largest percentage of persons of prime working age own their houses, and in which country are homeownership rates lowest?

**Question:** Are all the non-homeowners renters? If not, what other housing types are possible?

### **Guidelines**

- In order to merge two datasets both datasets must necessarily contain at least one “merge variable”, which exists in both files, indicating which household-level cases go with which person-level records. The variable links observations with the same value in the two datasets (in case of several “merge variables”, the observations are linked on the same combination of values). In LIS datasets, the variable **hid** uniquely identifies households, and can be found in both the person and household files.
- SAS reminder: When merging datasets – performing a *many-to-one* merge – in SAS, use the **MERGE** statement following by a **BY** statement. Prior to use it, both datasets must be sorted. To sort the dataset, apply the procedure shown below and then merge the household and individual files on **hid**:

```
PROC SORT DATA=dataset;  
  BY variable;  
RUN;  
  
DATA mergefile;  
  MERGE householdfile Individualfile;  
  BY hid;  
  ... yourcode...  
RUN;
```

Only keep the list of household-level and individual-level variables that you wish to keep in the merged file to avoid unnecessary burden on the machine so that submitted jobs will run faster.

- Remember that when recoding variables, you can find a listing of the possible values of the original variable in [METIS](#). In this case, go to the LIS Database information. Select DE04, GR04 and US04. Select the variable own. Go to Cross-compare and click on the variable name to see the statistics and labels of the variables
- Like the household file, the person file contains weight variables. These variables can be used to weight by person, as an alternative to the method of multiplying household weights by number of household members that was used in the Part I. Although home ownership is a household-level variable, you will want to use the person weight to determine the proportion of *persons* who live in owner-occupied dwellings. For now, use the variable **ppopwgt**, which inflates to the total population size.

## Program

```
OPTIONS NONOTES NOSOURCE NOFMterr NODATE NONUMBER NOCENTER LABEL LS=MAX PS=MAX ;
TITLE "";

%MACRO sample ;
    %LET i = 1 ;
    %DO %UNTIL (&i > 3) ;
        %LET ccyy = %SCAN(&all,&i) ;
        TITLE "Country : &ccyy" ;

        DATA &ccyy.h ;
        SET &&&ccyy.h (KEEP=hid own) ;
        RUN ;
        PROC SORT DATA=&ccyy.h ;
            BY hid ;
        RUN ;
        DATA &ccyy.p ;
        SET &&&ccyy.p (KEEP= hid dname pwgt ppopwgt relation partner children age sex
immigr educ
                                educ_c emp status1 ptime gross1);
        RUN ;
        PROC SORT DATA=&ccyy.p ;
            BY hid ;
        RUN ;

        DATA &ccyy ;
        MERGE &ccyy.h &ccyy.p ;
            BY hid ;
            IF ((25 <= age <= 54) AND (relation LE 2200)) ;
            IF (100 <= own <= 199) THEN homeowner = 1;
            IF (200 <= own <= 299) THEN homeowner = 0;

        RUN ;
        PROC MEANS DATA=&ccyy ;
            VAR homeowner ;
            WEIGHT ppopwgt ;
        RUN ;

        %LET i = %EVAL(&i+1) ;
    %END ;
%MEND sample ;

%LET all = us04 de04 gr04 ;
%sample
```

## **Results**

Dataset	Homeownership %
US04	71.5%
DE04	42.7%
GR04	67.6%

**Question:** In which country do the largest percentage of persons of prime working age own their houses, and in which country are homeownership rates lowest?

- Homeownership rates are highest in the United States at 71.5%, and lowest in Germany at only 40.9%.

**Question:** Are all the non-homeowners renters? If not, what other housing types are possible?

- No, in these three datasets there is also a category labeled "free housing". The LIS variable definitions available through [METIS](#) explains that this can include "housing provided by employer, government or others, or illegal occupation".

## **Comments**

- You will notice that in this exercise the merge worked perfectly, i.e. all observations of the merging file were uniquely linked to one observation in the using file. This is always the case with LIS household and individual level files from the same dataset because all individuals belong to at least one and no more than one household.

## 2. Stacking data, employment rates by gender

### Goal

So far, we have performed all the analysis separately for each dataset, working with only one country at a time. For this and all subsequent analyses, however, we will create a “stacked” dataset that contains information for all three countries in a single file. This means your dataset will have as many value observations as your countries altogether have stored within one file. This is necessary if you wish to analyse several countries simultaneously, which we will be doing in the regressions in exercise 2.8. In case of independent cross country comparisons, the use of a stacked dataset is an alternative technique to the looping through datasets.

After creating a combined dataset, we will examine rates of employment and part-time employment by gender, and see how it differs among these three countries. As in the previous exercise, we will be looking only at prime-aged persons who are defined as household heads or partners of household heads.

We will be using the LIS variable **emp**, an indicator that reports whether or not a person is currently employed. This variable will ideally be coded according to the International Labor Organization (ILO) definition of employment, which identifies persons working at a certain moment in time. By this definition, a person may be considered as employed as soon as he/she has carried out any work.

Rates of employment and full-time employment among prime-age men tend to be similar and consistently high across countries, so we will be paying particular attention to differences in employment outcomes among women.

### Activity

Modify your program so that it first creates a combined data file for the United States, Germany, and Greece, and then performs any necessary recoding and produces descriptive statistics.

Create a set of cross-tabulations that shows the rates of prime-age employment by gender within each country. Create another set of cross-tabulations showing the rates of part-time work by gender within each country, among those who are employed. Use your results to complete the following table below.

You should write your code so that your overall program is broken down into three subroutines. The first subroutine should contain only the code needed to create the merged, stacked dataset. The second subroutine should contain all of the data-preparation and recodings. The third subroutine should contain code that produces the summary statistics. Your overall program can then simply call these two subroutines to make the dataset and output the results. Breaking up your code in this way will be important for making the program compact and efficient in later exercises.

Dataset	Female employment rate	Part-time employment rate among employed women
US04		
DE04		
GR04		

**Question:** Contrast these countries in terms of their rates of female employment (high or low) and their rates of part-time employment among employed women (high or low).

### **Guidelines**

- You do not need to remove the code that you used to produce the homeownership variable in the last exercise since the exercises gradually introduce variables that we will be using in the rest of our analysis.
- To put all of your datasets together, you will need to append the datasets. Use the **PROC APPEND** procedure and use the option **FORCE** to avoid any **SAS format errors** generated if two different SAS formats are used for the same variables.

```
PROC APPEND BASE=mybase DATA=nextdataset FORCE;
RUN;
```

- Here is example code that will stack the three merged datasets previously created:

```
%MACRO stack ;
    %LET i = 1 ;
    %DO %UNTIL (&i > 3) ;
        %LET ccy = %SCAN(&all,&i) ;
```

*... Merge your files as in the previous exercise ...*

```
%IF %EVAL(&i) = 1 %THEN
  %DO ;
    DATA current ;
    SET &ccyy ;
    RUN ;
  %END ;
%ELSE
  %DO ;
    PROC APPEND BASE=current DATA=&ccyy FORCE ;
    RUN ;
  %END ;

  %LET i = %EVAL(&i+1) ;
%END ;
```

*... rest of your code ...*

```
%MEND stack ;
```

This example assumes that you have stored your list of datasets in the global macro **&all**, and that **us04** is the first dataset listed. When the code above processes United States 2004 data, it will create your base file. For subsequent countries, it will append the stacked file to the current dataset.

- Once you have produced the stacked data file, you can produce descriptives for multiple countries at once by using the **PROC FREQ** procedure (**See. guidelines PART I – exercise 2**)

You can get subset the original dataset for processing by the **PROC FREQ** procedure using a SAS **WHERE** clause, a logical expression that generally consists of a sequence of operands (e.g.: constants, variables, etc.) and operators (e.g.: = < etc.) This clause allows SAS to select observations from datasets that meet a particular condition. For instance, you can only apply your procedure on employed persons with this piece of code: **WHERE emp = 1 ;**

Here is example code that will create a set of cross-tabulations showing the rates of part-time work by gender

```
PROC FREQ DATA=current;
  WHERE emp = 1;
  TABLES dname*sex*ptime / NOFREQ NOCOL NOCUM;
  WEIGHT ppopwgt ;
RUN ;
```

## **Program**

```
OPTIONS NONOTES NOSOURCE NOFMterr NODATE NONUMBER NOCENTER LABEL LS=MAX PS=MAX ;
TITLE "";

%MACRO stack ;
    %LET i = 1 ;
    %DO %UNTIL (&i > 3) ;
        %LET ccyy = %SCAN(&all,&i) ;

        DATA &ccyy.h ;
        SET &&&ccyy.h (KEEP=hid own) ;
        RUN ;
        PROC SORT DATA=&ccyy.h ;
            BY hid ;
        RUN ;
        DATA &ccyy.p ;
        SET &&&ccyy.p (KEEP=hid dname pwgt ppopwgt relation partner ageyoch age sex immigr educ
educ_c
                                emp status1 ptime gross1) ;
        RUN ;
        PROC SORT DATA=&ccyy.p ;
            BY hid ;
        RUN ;
        DATA &ccyy ;
        MERGE &ccyy.h &ccyy.p ;
            BY hid ;
            IF ((25 <= age <= 54) AND (relation LE 2200)) ;
            IF (100 <= own <= 199) THEN homeowner = 1 ;
            IF (200 <= own <= 299) THEN homeowner = 0 ;
        RUN ;

        %IF %EVAL(&i) = 1 %THEN
            %DO ;
                DATA current ;
                SET &ccyy ;
                RUN ;
            %END ;
        %ELSE
            %DO ;
                PROC APPEND BASE=current DATA=&ccyy FORCE ;
                RUN ;
            %END ;

        %LET i = %EVAL(&i+1) ;
    %END ;

    PROC SORT DATA=current ;
        BY dname sex ;
    RUN ;
    PROC FREQ DATA=current ;
        TABLES dname*sex*emp / NOFREQ NOCOL NOCUM ;
        WEIGHT ppopwgt ;
    RUN ;
    PROC FREQ DATA=current ;
        WHERE emp = 1 ;
        TABLES dname*sex*ptime / NOFREQ NOCOL NOCUM ;
        WEIGHT ppopwgt ;
    RUN ;
%MEND stack ;

%LET all = us04 de04 gr04 ;
%stack
```

## **Results**

Dataset	Female employment rate	Part-time employment rate among employed women
US04	72.4%	19.6%
DE04	78.5%	43.3%
GR04	58.3%	15.8%

**Question:** Contrast these countries in terms of their rates of female employment (high or low) and their rates of part-time employment among employed women (high or low).

- Employment rates among prime age workers are relatively high in the United States in Germany, and lower in Greece. In the United States, most employed women work full time, while nearly half of employed German women work part time. Greece combines low employment rates with high rates of full time employment among those women who are employed.

## **Comments**

- While **emp** will ideally contain employment status based on the ILO definition, in some cases it must be based on usual or main activity status. See the dataset-specific notes to the variable for information.
- There are several other variables that measure employment status in LIS, which may be useful for different purposes. More detailed information on current labor force status is contained in **clfs**, which is usually the variable used to construct **emp**.
- Those interested in those for whom employment is the main activity should consider the main activity variables **mainemp** and **cmas**.
- Those wishing to perform income distribution analyses according to the activity status of individuals (e.g., calculation of poverty rates for the working poor), may choose to focus on employment over a longer reference period. In this case, you need to be able to identify an individual's primary activity over the income reference period (which in LIS is normally one year), using the LIS variable **umas**.

### 3. Family structure and employment

#### Goal

In the previous exercise we examined cross-national differences in women's employment. In this exercise, we will examine the variation in employment rates among women, based on their partnership and family status. We will contrast partnered and single women. Within each of those two categories, we will contrast women without children in the household, women with young children, and women with older children. The variables created in this exercise will be useful later, when we combine family structure with other personal characteristics in a multivariate analysis of wages.

#### Activity

Since you already created the merged, stacked dataset in the previous exercise, you do not need to create it again. Modify your code so that the subroutine that merges and stacks the data is commented out, and add a line that simply loads the merged and stacked file at the beginning of the program.

Create a variable **achildcat**, to indicate the age of the youngest own child living in the household. This variable should be equal to 0 if there are no children under 18, equal to 1 if the youngest child is under 6 years old, and equal to 2 if the youngest child is between 6 and 17. You can create this variable based on the information in the variable **children**.

Create another variable **livepartner**, to indicate whether a person is currently living with a partner (=1) or not (=0). We will not distinguish married couples from non-married couples for this analysis, and will simply consider a person partnered in either case. If someone has a partner but is not living with them, however, we will *not* consider them partnered (**livepartner**=0).

Using the new variables you have created and the employment indicator you used in the previous exercise, produce summary statistics to fill the table below.

## Employment Rates

Dataset	All women	Partnered			Single		
		No children under 18	Child under 6	Child 6-17	No children under 18	Child under 6	Child 6-17
US04							
DE04							
GR04							

**Question:** Within each country, which subpopulation of prime age women has the lowest employment rates?

### Guidelines

- Take all the guidelines from the previous exercise to complete activities requested in this one.
- You may find it helpful to attach **SAS formats** to your new variables in order to get more user-friendly outputs. To do so, use the **PROC FORMAT** procedure and apply the formats to the newly created variables. When you want to create a format that specifies character strings to use to print variable values, you need to use the statement **VALUES**, which allows the assignment of multiple labels to internal values (possible values of a SAS variable):

```
PROC FORMAT ;  
VALUE <name1>  
    Intval1 = "yourlabel1"  
    Intval2 = "yourlabel2"  
    ;  
VALUE <name2>  
    Intval1 = "yourlabel1"  
    Intval2 = "yourlabel2"  
    ...  
    ;  
RUN ;
```

To assign the created format to a variable during a **DATA STEP**, you need to use the Statement **ATTRIB**. Note that you must add a dot to the name of the format for SAS to identify the name as a **SAS FORMAT**.

```
DATA ... ;  
SET ... ;  
ATTRIB myvariable FORMAT=name. ;  
    ...  
RUN ;
```

- Here is an example of code that would allow you to create labels the variables **achildat** and **livepartner**:

```
PROC FORMAT ;  
VALUE chcat  
    0 = 'no children <18'  
    1 = '<6 years'  
    2 = '6-17 years'  
    . = 'missing'  
;  
VALUE lipar  
    0 = 'not living with partner'  
    1 = 'living with partner'  
    . = 'missing'  
;  
RUN ;  
  
DATA current ;  
    SET &ccyy ;  
    ATTRIB achildcat FORMAT=chcat. ;  
    ATTRIB livepartner FORMAT=lipar. ;  
RUN ;
```

## **Program**

```
OPTIONS NONOTES NOSOURCE NOFMterr NODATE NONUMBER NOCENTER LABEL LS=MAX PS=MAX
;
TITLE "";

%MACRO family ;

    %LET i = 1 ;
    %DO %UNTIL (&i > 3) ;
        %LET ccpy = %SCAN(&all,&i) ;

        DATA &ccpy.h ;
        SET &&&ccpy.h (KEEP=hid own) ;
        RUN ;
        PROC SORT DATA=&ccpy.h ;
            BY hid ;
        RUN ;
        DATA &ccpy.p ;
        SET &&&ccpy.p (KEEP=hid dname ppgwt ppopwgt relation partner children age sex
immigr
                                educ educ_c emp status1 ptime gross1) ;

        RUN ;
        PROC SORT DATA=&ccpy.p ;
            BY hid ;
        RUN ;
        DATA &ccpy ;
        MERGE &ccpy.h &ccpy.p ;
            BY hid ;
            IF ((25 <= age <= 54) AND (relation LE 2200)) ;
            IF (100 <= own <= 199) THEN homeowner = 1 ;
            IF (200 <= own <= 299) THEN homeowner = 0 ;
            IF (children in (140 200)) THEN achildcat = 0 ;
            ELSE IF (children = 110) THEN achildcat = 1 ;
            ELSE IF (children in (120 130)) THEN achildcat = 2 ;
            IF ((120 <= partner <= 200) | partner=100) THEN livepartner
= 0 ;
                                ELSE IF (partner = 110) THEN
livepartner = 1 ;
                                ELSE
livepartner = . ;
        RUN ;

        %IF %EVAL(&i) = 1 %THEN
            %DO ;
                DATA current ;
                SET &ccpy ;
                ATTRIB achildcat FORMAT=chcat. ;
                ATTRIB livepartner FORMAT=lipar. ;
                RUN ;
            %END ;
        %ELSE
            %DO ;
```

```

                PROC APPEND BASE=current DATA=&ccyy FORCE ;
                RUN ;
            %END ;

            %LET i = %EVAL(&i+1) ;
        %END ;

        PROC SORT DATA=current ;
            BY dname sex ;
        RUN ;

        PROC FREQ DATA=current ;
            TABLES dname*sex*livepartner*achildcat*emp / NOFREQ NOCOL NOCUM ;
            WEIGHT ppopwgt ;
        RUN ;

    %MEND family ;

    PROC FORMAT ;
        VALUE chcat
            0 = 'no children <18'
            1 = '<6 years'
            2 = '6-17 years'
            . = 'missing'
        ;
        VALUE lipar
            0 = 'not living with partner'
            1 = 'living with partner'
            . = 'missing'
        ;
    RUN ;

    %LET all = us04 de04 gr04 ;
    %family

```

## **Results**

### **Employment Rates**

Dataset	All women	Partnered			Single		
		No children under 18	Child under 6	Child 6-17	No children under 18	Child under 6	Child 6-17
US04	72.4%	75.8%	60.0%	73.5%	83.3%	70.5%	77.2%
DE04	75.2%	79.6%	80.1%	74.1%	86.3%	48.2%	77.7%
GR04	58.3%	50.6%	58.3%	61.4%	69.1%	64.2%	81.4%

**Question:** Within each country, which subpopulation of prime age women has the lowest employment rates?

- In the United States, partnered women with young children have the lowest employment rates. In Greece, partnered women without children have the lowest employment rates. In Germany, however, single mothers with children have lower employment rates. This may be because of more generous child policy in Germany that makes it easier for mothers of young children to support themselves without paid employment.

### **Comments**

- There is no clear-cut definition of a single-mother household. In this exercise, we allow other adult members to be present (as long as they are not defined as her partner). An alternative approach would be to limit the sample to households composed of a single female adult and her children. Another possibility is to limit single mother households to those with children under a specified age limit.
- When subdividing subsets of the population as has been done here, pay attention to sample sizes. In small datasets, estimates for narrowly defined groups may become very small, making estimates less reliable. The estimate for single Greek women with young children in this exercise, for example, is based on only 17 cases!

## 4. Dependent employment and hourly wages

### Goal

In the next several exercises, we will shift from considering employment to analysing the earnings of those who are employed. We will focus our analysis on the hourly wages and thus restrict our sample to those in *dependent* employment only — that is, those who are employees. The self-employed, along with several other small categories of workers, are excluded.

In this exercise we will first determine how many workers are excluded from the analysis when the sample is restricted to those in dependent employment. We will then measure the gap in hourly wages for men and women, in each of the three countries in our study.

We will be using a measure of gross hourly wages, which is available in the three datasets we are using. In other datasets, however, it could be that only annual wages were available. In such cases, researchers must take care to account for variations in employment over the year, perhaps by restricting the sample to full-year, full-time workers.

In part I, we have introduced bottom- and topcoding as a technique to deal with extreme values. In this exercise we will also introduce a different technique that deals with extreme values: ‘trimming’ the distribution (i.e. deleting rather than recoding the extreme values). This technique is especially useful when calculating measures that are not defined for non positive values (such as logarithmic measures). In later exercises we will convert the hourly wages into logs, and thus we need to make sure that the sample that we analyse at this stage is the same that we will keep for our final analysis.

### Activity

Recode the variable **status1** to create a new variable **depemp** that indicates whether a person is in dependent employment. Using this variable, produce summary statistics reporting the proportion of dependent employment among prime-age male workers and among prime-age female workers, and fill in the following table.

## Employment Rates

Dataset	Men		Women	
	Dependent Employment (%)	Non-dependent employment (%)	Dependent Employment (%)	Non-dependent employment (%)
US04				
DE04				
GR04				

Next, use the LIS hourly wage variable **gross1** to construct a new hourly wage variable **hourwage**, where the bottom and the top of the distribution are corrected as follows:

- at the top, we will carry out the same topcode as used in Part I: hourly wages greater than ten times the national median are set to ten times the national median (make sure you calculate the median and apply the topcode *separately* for each country);
- at the bottom, we will 'trim' the distribution so that observations with negative and zero hourly wages are set to missing.

Using this new hourly wage variable, calculate the *gender wage gap* for dependent employees in each country. The gender wage gap is defined here as the ratio of the median wages of women to the median wages of men. Use your results to complete the table below.

	Gender wage gap for dependent employees
US04	
DE04	
GR04	

**Question:** Does the percentage of workers not in dependent employment differ substantially across countries? Does it differ between men and women?

**Question:** Which country has the most wage inequality between men and women, among dependent employees?

## **Guidelines**

- For the purpose of topcoding, you can adjust the code already used in previous exercises. (see Part I, exercise 3). Here is an example of code to do it:

```
PROC MEANS DATA=&ccyy NOPRINT ;  
VAR hourwage ;  
WEIGHT ppopwgt ;  
OUTPUT OUT=tmp MEDIAN=med;  
RUN ;  
DATA _NULL_;  
SET tmp;  
CALL SYMPUT("wins",med);  
RUN;  
DATA &ccyy ;  
SET &ccyy ;  
IF hourwage gt (&wins * 10) THEN hourwage = &wins * 10 ;  
RUN ;
```

- As the **PROC FREQ** procedure allows you to create a set of cross-tabulations for discrete variables, the **PROC MEANS** procedure helps you to produce set of cross-tabulations for continuous variables. You then need to use the following statement:

**CLASS** Identify variables whose values define subgroups for the analysis  
**TYPES** Identify specific combinations of class variables to use to subdivide the data

As an example the code below calculates the median earnings cross-tabulated by gender for each country:

```
PROC MEANS DATA=current MEDIAN;  
CLASS dname sex;  
TYPES dname*sex;  
VAR hourwage ;  
WEIGHT ppopwgt ;  
RUN ;
```

## **Program**

```
OPTIONS NONOTES NOSOURCE NOFMterr NODATE NONUMBER NOCENTER LABEL LS=MAX PS=MAX
;
TITLE "";

%MACRO depend ;

    %LET i = 1 ;
    %DO %UNTIL (&i > 3) ;
        %LET ccyy = %SCAN(&all,&i) ;

        DATA &ccyy.h ;
            SET &&&ccyy.h (KEEP=hid own) ;
        RUN ;
        PROC SORT DATA=&ccyy.h ;
            BY hid ;
        RUN ;
        DATA &ccyy.p ;
            SET &&&ccyy.p (KEEP=hid dname pwgt ppopwgt relation partner children age sex
immigr
                                educ educ_c emp status1 ptime gross1) ;

        RUN ;
        PROC SORT DATA=&ccyy.p ;
            BY hid ;
        RUN ;

        DATA &ccyy ;
            MERGE &ccyy.h &ccyy.p ;
                BY hid ;
                    IF ((25 <= age <= 54) AND (relation LE 2200)) ;
                    IF (100 <= own <= 199) THEN homeowner = 1 ;
                    IF (200 <= own <= 299) THEN homeowner = 0 ;

                    IF      (children in (140 200)) THEN achildcat = 0 ;
                    ELSE IF (children = 110)           THEN achildcat = 1 ;
                    ELSE IF (children in (120 130)) THEN achildcat = 2 ;

                    IF ((120 <= partner <= 200) | partner = 100) THEN livepartner = 0 ;
                    ELSE IF (partner = 110)           THEN livepartner =
1 ;
                                ELSE
                                livepartner
= . ;

                    IF      (100 <= status1 <= 122) THEN depemp = 1 ;
                    ELSE IF (200 <= status1 <= 240) THEN depemp = 0 ;
                    ELSE
                                depemp = . ;

                    hourwage = gross1 ;
                    IF      (hourwage <= 0 ) THEN hourwage = . ;

        RUN ;

        PROC MEANS DATA=&ccyy NOPRINT ;
            VAR hourwage ;
```

```

        WEIGHT ppopwgt ;
        OUTPUT OUT=tmp MEDIAN=med;
        RUN ;
        DATA _NULL_;
        SET tmp;
        CALL SYMPUT("wins",med);
        RUN;
        DATA &ccyy ;
        SET &ccyy ;
        IF hourwage gt (&wins * 10) THEN hourwage = &wins * 10 ;
        RUN ;

        %IF %EVAL(&i) = 1 %THEN
            %DO ;
                DATA current ;
                SET &ccyy ;
                ATTRIB achildcat FORMAT=chcat. ;
                ATTRIB livepartner FORMAT=lipar. ;
                RUN ;
            %END ;
        %ELSE
            %DO ;
                PROC APPEND BASE=current DATA=&ccyy FORCE ;
                RUN ;
            %END ;

            %LET i = %EVAL(&i+1) ;
        %END ;

        PROC SORT DATA=current ;
        BY dname sex ;
        RUN ;

        PROC FREQ DATA=current ;
        WHERE emp = 1 ;
        TABLES dname*sex*depemp / NOFREQ NOCOL NOCUM ;
        WEIGHT ppopwgt ;
        RUN ;

        PROC MEANS DATA=current MEDIAN;
        CLASS  dname sex;
        TYPES dname*sex ;
        VAR hourwage ;
        WEIGHT ppopwgt ;
        RUN ;

        %MEND depend ;

        PROC FORMAT ;
        VALUE chcat
            0 = 'no children <18'
            1 = '<6 years'
            2 = '6-17 years'
            . = 'missing'

```

```
;
VALUE lipar
    0 = 'not living with partner'
    1 = 'living with partner'
    . = 'missing'
;
RUN ;

%LET all = us04 de04 gr04 ;
%depend
```

## **Results**

### **Employment Rates**

Dataset	Men		Women	
	Dependent Employment (%)	Non-dependent employment (%)	Dependent Employment (%)	Non-dependent employment (%)
US04	86%	14%	91%	9%
DE04	87%	13%	92%	8%
GR04	66%	34%	70%	30%

### **Gender wage gaps**

	Gender wage gap for dependent employees
US04	0.75
DE04	0.82
GR04	0.85

**Question:** Does the percentage of workers not in dependent employment differ substantially across countries? Does it differ between men and women?

- Greece has a much higher rate of non-dependent employment (which is primarily self-employment). In all the countries, women have somewhat higher rates of dependent employment than men. Keep in mind, therefore, that the results in the subsequent exercises may be somewhat unrepresentative, particularly for Greece, because they exclude a substantial number of workers.

**Question:** Which country has the most wage inequality between men and women, among dependent employees?

- The United States shows the largest gender wage gap. Among prime age workers, the median hourly wage of women is only 75% that of men.

## **Comments**

- The hourly wage variable **gross1** reports the wage in the first job. If information is available for a second job, it will be in the variable **gross2**. All other job characteristics are also organised in this way (e.g. industry, occupation, etc.).
- The wage gap calculated here is based on the median, but some researchers calculate an alternative version based on the mean, which will give slightly different results.

## 5. Hourly wages, education, and country-specific variables

### Goal

This exercise continues the analysis of gender wage gaps in hourly wages among those in dependent employment, which we started to program in the previous exercise. In this exercise, we will see how gender disparities in wages differ by educational attainment.

This exercise also demonstrates the use of two different LIS variables measuring educational attainment. One is fully standardised for cross-national compatibility, but contains few categories. The other may contain more information, but has country-specific codes, and thus requires researchers to perform their own standardisation.

The standardised variable is called **educ**, which is based on the International Standard Classification of Education (ISCED). The non-standardised version (from which **educ** is constructed) is **educ\_c**. This is one of many attributes for which LIS provides both a standardised and country-specific variable. Any variable ending in **\_c** is non-standardised, meaning that it can have different contents in different datasets. It is important to carefully examine the dataset-specific documentation before using such variables.

### Activity

Add code to your program to create a table cross-tabulating the variables **educ** and **educ\_c** for each country. This will show how the standardised variable was constructed in each case. Be sure to:

- include missing values in your table, so that you can see whether any of the cases in the original education variable could not be allocated to a category in the standardised version;
- remove the value labels from the tabulation of the **educ\_c** variable (since the value labels of the **\_c** variables are by definition dataset-specific, in a stacked dataset with observations from several LIS datasets, the value labels of those variables will be incorrect, as they can only refer to one specific LIS dataset – usually the last one that was

used to construct the stacked data, see further details in the comments section of this exercise).

Using the hourly wage variable **hourwage** that you created in the last exercise, calculate the gender wage gap by education for each country, and complete the table below. The gender wage gap is defined as it was in the previous exercise. To obtain the earnings ratio by education, simply calculate the ratio separately for individuals in each of the three categories of the standardised education variable.

### Gender wage gaps by educational attainment

	Low education	Medium education	High education
US04			
DE04			
GR04			

**Question:** For each of the three countries, what are the categories in the original dataset that are recoded as “high education” in the standardised education variable?

**Question:** Are there any categories in the original **educ\_c** variable that could not be translated into the standardised form?

**Question:** In general, which educational attainment category shows the greatest earnings inequality between genders? How do the patterns differ by country?

### Guidelines

- SAS reminder: to remove the value labels from a variable apply this code when you prepare the stacked datasets:  
**FORMAT educ\_c ;**
- If you are uncertain about the meaning of a code in a categorical variable, consult [METIS](#) for variable information and codebook.

## Program

```
OPTIONS NONOTES NOSOURCE NOFMterr NODATE NONUMBER NOCENTER LABEL LS=MAX
PS=MAX ;
TITLE "";

%MACRO educ ;

    %LET i = 1 ;
    %DO %UNTIL (&i > 3) ;
        %LET ccyy = %SCAN(&all,&i) ;

        DATA &ccyy.h ;
            SET &&&ccyy.h (KEEP=hid own) ;
        RUN ;
        PROC SORT DATA=&ccyy.h ;
            BY hid ;
        RUN ;
        DATA &ccyy.p ;
            SET &&&ccyy.p (KEEP=hid dname pwgt ppopwgt relation partner children age sex
                immigr educ educ_c emp status1 ptime gross1) ;

        RUN ;
        PROC SORT DATA=&ccyy.p ;
            BY hid ;
        RUN ;

        DATA &ccyy ;
            MERGE &ccyy.h &ccyy.p ;
            BY hid ;
            IF ((25 <= age <= 54) AND (relation LE 2200)) ;
            IF (100 <= own <= 199) THEN homeowner = 1 ;
            IF (200 <= own <= 299) THEN homeowner = 0 ;
            IF (children in (140 200)) THEN achildcat = 0 ;
            ELSE IF (children = 110) THEN achildcat = 1 ;
            ELSE IF (children in (120 130)) THEN achildcat = 2 ;
            IF ((120 <= partner <= 200) | partner = 100) THEN livepartner = 0 ;
            ELSE IF (partner = 110) THEN livepartner = 1 ;
            ELSE livepartner = . ;

;

            IF (100 <= status1 <= 122) THEN depemp = 1 ;
            ELSE IF (200 <= status1 <= 240) THEN depemp = 0 ;
            ELSE depemp = . ;
            hourwage = gross1 ;
            IF (hourwage <= 0 ) THEN hourwage = . ;
            FORMAT educ_c ;
        RUN ;

        PROC MEANS DATA=&ccyy NOPRINT ;
            VAR hourwage ;
            WEIGHT ppopwgt ;
            OUTPUT OUT=tmp MEDIAN=med;
        RUN ;
        DATA _NULL_ ;
            SET tmp ;
            CALL SYMPUT("wins",med);
        RUN ;
        DATA &ccyy ;
```

```

SET &ccyy ;
  IF hourwage gt (&wins * 10) THEN hourwage = &wins * 10 ;
RUN ;

%IF %EVAL(&i) = 1 %THEN
  %DO ;
    DATA current ;
    SET &ccyy ;
    ATTRIB achildcat FORMAT=chcat. ;
    ATTRIB livepartner FORMAT=lipar. ;
  RUN ;
  %END ;
%ELSE
  %DO ;
    PROC APPEND BASE=current DATA=&ccyy FORCE ;
    RUN ;
  %END ;

  %LET i = %EVAL(&i+1) ;
%END ;

PROC SORT DATA=current ;
  BY dname sex ;
RUN ;

PROC FREQ DATA=current ;
  TABLES dname*educ*educ_c / NOFREQ NOCOL NOCUM MISSING ;
RUN ;

PROC MEANS DATA=current MEDIAN;
  CLASS dname sex educ ;
  TYPES dname*sex*educ ;
  VAR hourwage ;
  WEIGHT ppopwgt ;
  RUN ;

%MEND educ ;

PROC FORMAT ;
  VALUE chcat
    0 = 'no children <18'
    1 = '<6 years'
    2 = '6-17 years'
    . = 'missing'
  ;
  VALUE lipar
    0 = 'not living with partner'
    1 = 'living with partner'
    . = 'missing'
  ;
RUN ;

%LET all = us04 de04 gr04 ;
%educ

```

## **Results**

### **Gender wage gaps by educational attainment**

	Low education	Medium education	High education
US04	0.71	0.71	0.78
DE04	0.78	0.83	0.81
GR04	0.67	0.78	0.96

**Question:** For each of the three countries, what are the categories in the original dataset that are recoded as “high education” in the standardised education variable?

- In the United States, high education combines those with associate degrees, bachelor's degrees, and advanced degrees (masters, professional school, or doctorate).
- In Germany, high education combines those classified as having “higher vocational” (of any kind) or “higher education”.
- In Greece, high education includes those with tertiary graduate or postgraduate level education.

**Question:** Are there any categories in the original **educ\_c** variable that could not be translated into the standardised form?

- In the United States, all values of **educ\_c** receive a value in **educ**. In Germany, a small number of persons categorised as “inadequately, other diploma” or “still in education” are set to missing. In Greece, a small number of persons listed as “still in education” are set to missing.

**Question:** In general, which educational attainment category shows the greatest earnings inequality between genders? How do the patterns differ by country?

- In all three countries, there is a smaller gender wage gap among highly educated workers. This is particularly notable in Greece. In that country, wage inequality is greater among the low-educated than in the United States and Germany, but there is near equality among the highly educated.

## **Comments**

You may have some doubts and questions why we did advise you in the exercise to not show the labels of **educ\_c** for the cross-tabulation of **educ** and **educ\_c**. As you are aware we did append values for several countries for each variable to get the stacked file. In your stacked file, for standardised variables these values have all the same meaning, as the values and labels are completely the standardised. However, it is more complicated for non-standardized values and labels of **\_c** variables, as each dataset has its own values and own meaning, as indicated by the labels attached to the data.

Be aware that while appending the data, your programming software will very likely overwrite the label automatically. Thus we do in general advise you to drop the labels from the variables **\_c** within your code. You always have the full information on the labels in the codebooks. However, if you prefer to keep the full labels somewhere in the data there are several solutions.

The simple solution is that you tabulate each country separately (see for example exercise 2 of part II) before you generate the stacked version. As a second solution, you can also easily rename the variables **\_c** to the specific **\_`ccyy'** of each dataset - this way you will append a separate variable **\_`ccyy'** for each of your datasets, which does have only observations for the specific **`ccyy'** with the country specific labels attached. Be aware that you then need to tabulate for each **`ccyy'** separately to get the right percentage of missings!

# 6. Immigration and wages, understanding harmonisation

## Goal

Each of the countries we are examining has a significant immigrant population, and their labor market outcomes are often very different from those of the non-immigrant population. In this exercise we will compare the wages of immigrants and non-immigrant men and women, just as we compared individuals of different educational levels in the last exercise.

LIS provides a variable indicating whether someone is an immigrant, which we will be using in this exercise. However, the choices that go into constructing this variable are complex, because the information available to construct it varies widely from country to country. It is important to understand the assumptions behind variables such as this one, because in some cases researchers may prefer to develop their own standardisation procedures based on their particular needs.

## Activity

Using the top and bottom coded hourly wage variable, calculate the gender earnings ratio by immigration status for each country, and complete the table below. The gender earnings ratio is computed just as in the previous exercise, except that you will now subdivide the population into immigrants and non-immigrants, rather than by educational attainment categories.

### **Gender earnings ratios by immigration status**

	Non-immigrants	Immigrants
US04		
DE04		
GR04		

**Question:** What information is used to construct the **immigr** variable? If you wanted to determine how the indicator was constructed in a particular dataset, what other variables would you need to look at?

**Question:** Is gender earnings inequality larger among immigrants or non-immigrants? Does this differ by country?

## **Program**

```
OPTIONS NONOTES NOSOURCE NOFMterr NODATE NONUMBER NOCENTER LABEL LS=MAX PS=MAX
;
TITLE "";

%MACRO immig ;
    %LET i = 1 ;
    %DO %UNTIL (&i > 3) ;
        %LET cyy = %SCAN(&all,&i) ;

        DATA &cyy.h ;
            SET &&&cyy.h (KEEP=hid own) ;
        RUN ;
        PROC SORT DATA=&cyy.h ;
            BY hid ;
        RUN ;
        DATA &cyy.p ;
            SET &&&cyy.p (KEEP=hid dname pwgt ppopwgt relation partner children age sex
immigr                                educ educ_c emp status1 ptime gross1) ;
        RUN ;
        PROC SORT DATA=&cyy.p ;
            BY hid ;
        RUN ;

        DATA &cyy ;
            MERGE &cyy.h &cyy.p ;
                BY hid ;
                    IF ((25 <= age <= 54) AND (relation LE 2200)) ;
                    IF (100 <= own <= 199) THEN homeowner = 1 ;
                    IF (200 <= own <= 299) THEN homeowner = 0 ;
                    IF (children in (140 200)) THEN achildcat = 0 ;
                    ELSE IF (children = 110) THEN achildcat = 1 ;
                    ELSE IF (children in (120 130)) THEN achildcat = 2 ;
                    IF ((120 <= partner <= 200) | partner = 100) THEN
livepartner=0;
                    ELSE IF (partner = 110) THEN
livepartner=1;
                    ELSE
                        livepartner=.;
                    FORMAT educ_c ;
                    IF (100 <= status1 <= 122) THEN depemp = 1 ;
                    ELSE IF (200 <= status1 <= 240) THEN depemp = 0 ;
                    ELSE
                        depemp = . ;
                    hourwage = gross1 ;
                    IF (hourwage <= 0) THEN hourwage = . ;
        RUN ;

        PROC MEANS DATA=&cyy NOPRINT ;
            VAR hourwage ;
            WEIGHT ppopwgt ;
            OUTPUT OUT=tmp MEDIAN=med;
        RUN ;
        DATA _NULL_ ;
            SET tmp;
            CALL SYMPUT("wins",med);
        RUN ;
    %END ;
%END ;
```

```

DATA &ccyy ;
  SET &ccyy ;
  IF hourwage gt (&wins * 10) THEN hourwage = &wins * 10 ;
RUN ;

%IF %EVAL(&i) = 1 %THEN
  %DO ;
    DATA current ;
    SET &ccyy ;
    ATTRIB achildcat FORMAT=chcat. ;
    ATTRIB livepartner FORMAT=lipar. ;
  RUN ;
%END ;
%ELSE
  %DO ;
    PROC APPEND BASE=current DATA=&ccyy FORCE ;
    RUN ;
  %END ;

  %LET i = %EVAL(&i+1) ;
%END ;

PROC SORT DATA=current ;
  BY dname sex ;
RUN ;

PROC MEANS DATA=current MEDIAN ;
CLASS  dname sex immigr ;
  TYPES  dname*sex*immigr ;
  VAR hourwage ;
  WEIGHT ppopwgt ;
RUN ;
%MEND immigr ;

PROC FORMAT ;
  VALUE chcat
    0 = 'no children <18'
    1 = '<6 years'
    2 = '6-17 years'
    . = 'missing'
  ;
  VALUE lipar
    0 = 'not living with partner'
    1 = 'living with partner'
    . = 'missing'
  ;
RUN ;

%LET all = us04 de04 gr04 ;
%immigr

```

## **Results**

### **Gender earnings ratios by immigration status**

	Non-immigrants	Immigrants
US04	0.76	0.82
DE04	0.80	0.77
GR04	0.89	0.82

**Question:** What information is used to construct the **immigr** variable? If you wanted to determine how the indicator was constructed in a particular dataset, what other variables would you need to look at?

- As indicated in the Variable Definition of **immigr** using [METIS](#) documentation tool, Immigrants are defined by LIS as all persons who have that country as country of usual residence and (in order of priority):
  - whom the data provider defined as immigrants;
  - who self-define them-selves as immigrants;
  - who are the citizen/national of another country;
  - who were born in another country.
- **immigr** could be constructed out of **citizen**, **ctrybrth**, **yrsresid**, **ethnic\_c** and **immigr\_c**.

**Question:** Is gender earnings inequality larger among immigrants or non-immigrants? Does this differ by country?

- In the United States, the gender wage gap is greater among non-immigrants, but in Germany and Greece it is greater among immigrants.

## **Comments**

- The definition of immigrant used in **immigr** may differ substantially from dataset to dataset. The variables that may be used in its construction include **citizen**, **ctrybrth**, **yrsresid**, **ethnic\_c** and **immigr\_c**.

## 7. Wage regressions

### Goal

We have seen how employment varies by gender and family structure, and how gender wage gaps vary by education and immigration status. In this exercise, we will investigate the impact of all these variables on wages, using a multivariate regression.

Wages are generally not normally distributed. We will therefore apply a logarithmic transformation in order to create an outcome variable that is approximately normal, which is more suitable for regression modeling.

In addition to the variables we have already seen, we will also control for age, which has a strong relationship with earnings. Since the relationship between age and income is not necessarily linear, we will also add a term for age-squared.

### Activity

Create a new variable, **logwage**, which is the natural logarithm of the **hourwage** variable you created earlier.

Create a variable **agesq**, which is equal to **age** squared.

Create dummy variables based on the **achildcat** and **educ** variables you have used previously. You will need the following variables:

- **youngchild**, equal to 1 if the person has a child under 6 in the household, 0 otherwise.
- **oldchild**, equal to 1 if the person has a child age 6-17 in the household, 0 otherwise.
- **hieduc**, equal to 1 if the person has high educational attainment, 0 otherwise.
- **mededuc**, equal to 1 if the person has medium educational attainment, 0 otherwise.

When including categorical variables in a regression, one category is always left out and used as the reference category. The reference categories in this case are having no non-adult children, and having low educational attainment.

After creating all of the variables, run regression models predicting **logwage**. You should run the regressions separately in each country, and within each

country you should run a separate model for men and women. Use the following model:

**logwage = f (age agesq mededuc hieduc immigr livepartner youngchild  
oldchild pttime homeowner)**

Produce a table of the six resulting models, with coefficients, standard errors, sample sizes, and r-squared values.

It is recommended that you create a new sub-routine to run the regressions, which can be called from your main program.

**Question:** Who receives a higher wage premium from being highly educated, men or women?

**Question:** When controlling for other individual characteristics, what is the relationship between immigrant status and wages?

**Question:** When controlling for other individual characteristics, do women with young children make more or less than women without children?

## **Guidelines**

➤ Linear regression in SAS is done with the **PROC REG** or **PROC SURVEYREG** procedures.

Use of the **PROC SURVEYREG** procedure is recommended to correct standard errors for possible heteroskedasticity.

```
PROC SURVEYREG;  
BY dname sex ;  
WEIGHT ppopwgt ;  
MODEL logwage = age agesq mededuc hieduc immigr livepartner youngchild  
oldchild pttime homeowner ;  
RUN;
```

**BY** : to obtain separate analyses on observations in groups defined by the BY variables.

**MODEL**: specifies the dependent (response) variable and the independent (regressor) variables or effects.

**WEIGHT** : names the variable that contains the sampling weights.

The code above estimates regressions for men in each country and then for women in each country.

## Program

```
%MACRO country ;

%LET i = 1 ;
%DO %UNTIL (&i > 3) ;
  %LET ccy = %SCAN(&all,&i) ;

  DATA &ccy.h ;
  SET &&&ccy.h (KEEP=hid own) ;
  RUN ;
  PROC SORT DATA=&ccy.h ;
  BY hid ;
  RUN ;
  DATA &ccy.p ;
  SET &&&ccy.p (KEEP=hid dname pwgt ppopwgt relation partner children age sex
              immigr educ educ_c emp status1 ptime gross1) ;
  RUN ;
  PROC SORT DATA=&ccy.p ;
  BY hid ;
  RUN ;

  DATA &ccy ;
  MERGE &ccy.h &ccy.p ;
  BY hid ;
  IF ((25 <= age <= 54) AND (relation LE 2200)) ;
  IF (100 <= own <= 199) THEN homeowner = 1 ;
  IF (200 <= own <= 299) THEN homeowner = 0 ;
  IF      (children in (140 200))      THEN achildcat = 0 ;
  ELSE IF (children = 110)              THEN achildcat = 1 ;
  ELSE IF (children in (120 130))      THEN achildcat = 2 ;
  IF      (achildcat = 1)              THEN youngchild = 1 ;
  ELSE IF (achildcat in (0,2))         THEN youngchild = 0 ;
  ELSE                                     youngchild = . ;
  IF      (achildcat = 2)              THEN oldchild = 1 ;
  ELSE IF (achildcat in (0,1))         THEN oldchild = 0 ;
  ELSE                                     oldchild = . ;
  IF ((120 <= partner <= 200) | partner = 100) THEN livepartner = 0 ;
  ELSE IF (partner = 110)              THEN livepartner = 1 ;
  ELSE                                  livepartner = . ;
  IF (100 <= status1 <= 122) THEN depemp = 1 ;
  ELSE IF (200 <= status1 <= 240) THEN depemp = 0 ;
  ELSE                                  depemp = . ;
  hourwage = gross1 ;
  IF (hourwage <= 0) THEN hourwage = . ;
  agesq = age*age ;

  FORMAT educ_c ;
  IF      (educ in (1,3))      THEN mededuc = 0 ;
  ELSE IF (educ = 2)          THEN mededuc = 1 ;
  ELSE                                  mededuc = . ;
  IF      (educ in (1,2)) THEN hieduc = 0 ;
  ELSE IF (educ = 3)          THEN hieduc = 1 ;
  ELSE                                  hieduc = . ;
  RUN ;

  PROC MEANS DATA=&ccy NOPRINT ;
  VAR hourwage ;
  WEIGHT ppopwgt ;
  OUTPUT OUT=tmp MEDIAN=med;
```

```

RUN ;
DATA _NULL_;
SET tmp;
CALL SYMPUT("wins",med);
RUN;
DATA &ccyy ;
SET &ccyy ;
IF hourwage gt (&wins * 10) THEN hourwage = &wins * 10 ;
logwage = LOG(hourwage) ;

RUN ;

%IF %EVAL(&i) = 1 %THEN
%DO ;
DATA current ;
SET &ccyy ;
ATTRIB achildcat FORMAT=chcat. ;
ATTRIB livepartner FORMAT=lipar. ;
RUN ;
%END ;
%ELSE
%DO ;
PROC APPEND BASE=current DATA=&ccyy FORCE ;
RUN ;
%END ;

%LET i = %EVAL(&i+1) ;
%END ;

PROC SORT DATA=current ;
BY dname sex ;
RUN ;

PROC SURVEYREG;
BY dname sex ;
WEIGHT ppopwgt ;
MODEL logwage = age agesq mededuc hieduc immigr livepartner youngchild oldchild
ptime homeowner ;
RUN;

%MEND country ;

PROC FORMAT ;
VALUE chcat
0 = 'no children <18'
1 = '<6 years'
2 = '6-17 years'
. = 'missing'
;
VALUE lipar
0 = 'not living with partner'
1 = 'living with partner'
. = 'missing'
;
RUN ;

%LET all = us04 de04 gr04 ;
%country

```

## **Results**

### **Men**

	us04	de04	gr04
age	0.050*** (0.0056)	0.053** (0.015)	0.036* (0.017)
agesq	-0.00049*** (0.000070)	-0.00059** (0.00019)	-0.00026 (0.00021)
mededuc	0.28*** (0.015)	0.012 (0.068)	0.16*** (0.026)
hieduc	0.68*** (0.016)	0.40*** (0.070)	0.48*** (0.035)
immigr	-0.058*** (0.012)	-0.066 (0.029)	-0.25*** (0.046)
livepartner	0.041*** (0.013)	0.024 (0.036)	-0.026 (0.043)
youngchild	0.062*** (0.012)	0.072 (0.034)	0.13*** (0.035)
oldchild	0.069*** (0.011)	0.048* (0.025)	0.11** (0.033)
ptime	-0.30*** (0.031)	-0.32* (0.10)	-0.27* (0.11)
homeowner	0.24*** (0.012)	0.11*** (0.023)	0.034 (0.027)
_cons	1.11*** (0.11)	1.24*** (0.31)	0.75* (0.33)
N	29986	3645	1147
R-sq	0.223	0.168	0.333

Standard errors in parentheses

\* p<0.05, \*\* p<0.01, \*\*\* p<0.001

**Women**

	us04	de04	gr04
age	0.042*** (0.0059)	0.052 (0.017)	0.069** (0.021)
agesq	-0.00044*** (0.000074)	-0.00059 (0.00021)	-0.00063* (0.00027)
mededuc	0.32*** (0.018)	0.083* (0.039)	0.34*** (0.038)
hieduc	0.75*** (0.018)	0.38*** (0.042)	0.78*** (0.041)
immigr	-0.018* (0.013)	-0.069 * (0.029)	-0.22*** (0.053)
livepartner	-0.012 (0.011)	-0.018 (0.032)	-0.050 (0.042)
youngchild	0.039** (0.013)	-0.0080 (0.045)	0.087* (0.044)
oldchild	-0.048*** (0.010)	-0.12* (0.031)	0.048 (0.037)
ptime	-0.17*** (0.012)	-0.091*** (0.027)	-0.099* (0.048)
homeowner	0.19*** (0.011)	0.073** (0.026)	-0.029 (0.033)
_cons	1.11*** (0.11)	1.22*** (0.33)	-0.21 (0.39)
N	28597	3562	963
R-sq	0.190	0.118	0.435

Standard errors in parentheses

\* p&lt;0.05, \*\* p&lt;0.01, \*\*\* p&lt;0.001

**Question:** Who receives a higher wage premium from being highly educated, men or women?

- In all three countries, the coefficient for high education is higher for women, indicating a larger wage premium from having high educational attainment.

**Question:** When controlling for other individual characteristics, what is the relationship between immigrant status and wages?

- In general, the association between immigrant status and wages is negative, although the 95% confidence interval for the coefficient crosses 0 for German men. In general, the negative association appears stronger in Germany than in the United States, and stronger in Greece than Germany.

**Question:** When controlling for other individual characteristics, do women with young children make more or less than women without children?

- In general, women with young children have higher wages than women without children in the United States and Greece, but there is no association in Germany. Higher wages for women with young children could be due to a selection effect, where mothers of young children are more likely to enter the labor market if they have higher earning power.

## **Comments**

- As we have seen, employment rates, particularly among women, vary substantially across countries. The wage regressions shown here do not account for this differential selection into employment. For this reason, many studies of wages apply a technique such as a Heckman correction, which attempts to correct for this selection bias.

## 8. Pooled regressions and normalised weights

### Goal

In the previous exercise, we ran parallel, separate regressions for each country. In this exercise, we see an alternative approach, in which all countries are pooled together in a single model. We will continue to use classical OLS regression, but the approach shown here can easily be extended for more complex multilevel estimation approaches.

The income variables in these datasets use different currencies. To compare them, we need to convert them to a common scale. We will apply *Purchasing Power Parity* (PPP) deflators, which are intended to ensure that equal quantities of income correspond to equivalent purchasing power across currencies and national economies. The PPP deflators provided here are taken from the Penn World Tables, which are a commonly used source.

Up to this point, we have been using the weight variable **ppopwgt**, which inflates to the total population. If we use this weight in a pooled regression, every household will receive equal weight. However, this would mean that Greece — which has a much smaller population than the United States or Germany — will not have much influence on the results. In order to give each country equal weight in the model, we will use the alternative normalized weight variable **pwgt**, which always sums to 10,000 within each dataset.

### Activity

Create a new logged hourly wage variable, **logwage\_ppp** by dividing **hourwage** by the following deflators (<http://www.lisdatacenter.org/data-access/web-tabulator/methods/ppp/>) and then taking the natural logarithm:

dataset	PPP deflator
US04	1
DE04	0.74
GR04	0.62

Create dummy variables for **germany** (=1 if the case is from German dataset, otherwise 0) and **greece** (=1 if the case is from the Greek dataset, otherwise 0). The United States will be the reference category.

Estimate the following model, for men and women separately:

**logwage = f (age agesq mededuc hieduc immigr livepartner youngchild oldchild ptime homeowner germany greece)**

Make sure to use *normalised*, not inflated weights.

Produce a table of the two resulting models, with coefficients, standard errors, sample sizes, and r-squared values.

**Question:** How can you interpret the meaning of the coefficients for the Germany and Greece dummy variables?

**Question:** In this pooled model, which carries a higher wage penalty: being an immigrant, or working part time?

## **Guidelines**

- Run your regressions as you did in the previous exercise. This time, make sure to
  - use *normalized* (**pwgt**) not inflated weights (**ppopwgt**)
  - sort your dataset on which you run the regression only by **sex** and not by **dname sex** since all countries are pooled together in a single model.

## Program

```
OPTIONS NONOTES NOSOURCE NOFMterr NODATE NONUMBER NOCENTER LABEL LS=MAX PS=MAX
MPRINT MLOGIC SYMBOLGEN;
TITLE "";
```

```
%MACRO pool ;
```

```
%LET i = 1 ;
```

```
%DO %UNTIL (&i > 3) ;
```

```
  %LET ccyy = %SCAN(&all,&i) ;
```

```
  DATA &ccyy.h ;
```

```
  SET &&&ccyy.h (KEEP=hid own) ;
```

```
  RUN ;
```

```
  PROC SORT DATA=&ccyy.h ;
```

```
    BY hid ;
```

```
  RUN ;
```

```
  DATA &ccyy.p ;
```

```
  SET &&&ccyy.p (KEEP=hid dname pwgt ppopwgt relation partner children age sex immigr
                educ educ_c emp status1 ptime gross1);
```

```
  RUN ;
```

```
  PROC SORT DATA=&ccyy.p ;
```

```
    BY hid ;
```

```
  RUN ;
```

```
    DATA &ccyy ;
```

```
    MERGE &ccyy.h &ccyy.p ;
```

```
    BY hid ;
```

```
    IF ((25 <= age <= 54) AND (relation LE 2200)) ;
```

```
    IF (100 <= own <= 199) THEN homeowner = 1 ;
```

```
    IF (200 <= own <= 299) THEN homeowner = 0 ;
```

```
    IF (children in (140 200)) THEN achildcat = 0 ;
```

```
    ELSE IF (children = 110) THEN achildcat = 1 ;
```

```
    ELSE IF (children in (120 130)) THEN achildcat = 2 ;
```

```
    IF (achildcat = 1) THEN youngchild = 1 ;
```

```
    ELSE IF (achildcat in (0,2)) THEN youngchild = 0 ;
```

```
    ELSE youngchild = . ;
```

```
    IF (achildcat = 2) THEN oldchild = 1 ;
```

```
    ELSE IF (achildcat in (0,1)) THEN oldchild = 0 ;
```

```
    ELSE oldchild = . ;
```

```
    IF ((120 <= partner <= 200) | partner = 100) THEN livepartner = 0 ;
```

```
    ELSE IF (partner = 110) THEN livepartner = 1 ;
```

```
    ELSE livepartner = . ;
```

```
      IF (100 <= status1 <= 122) THEN depemp = 1 ;
```

```
      ELSE IF (200 <= status1 <= 240) THEN depemp = 0 ;
```

```
      ELSE depemp = . ;
```

```
    hourwage = gross1 ;
```

```
    IF (hourwage <= 0) THEN hourwage = . ;
```

```
    agesq = age*age ;
```

```
    IF (educ in (1,3)) THEN mededuc = 0 ;
```

```
    ELSE IF (educ = 2) THEN mededuc = 1 ;
```

```
    ELSE mededuc = . ;
```

```
    IF (educ in (1,2)) THEN hieduc = 0 ;
```

```
    ELSE IF (educ = 3) THEN hieduc = 1 ;
```

```
    ELSE hieduc = . ;
```

```
    germany=0;
```

```
    IF dname = 'de04' THEN germany=1;
```

```
    greece =0;
```

```

    IF dname = 'gr04' THEN greece=1 ;
RUN ;

PROC MEANS DATA=&ccyy NOPRINT ;
  VAR hourwage ;
  WEIGHT ppopwgt ;
  OUTPUT OUT=tmp MEDIAN=med;
RUN ;
DATA _NULL_;
  SET tmp;
  CALL SYMPUT("wins",med);
RUN;
DATA &ccyy ;
  SET &ccyy ;
  IF hourwage gt (&wins * 10) THEN hourwage = &wins * 10 ;
  IF dname = 'de04' THEN ppp=0.74 ;
  ELSE IF dname = 'gr04' THEN ppp=0.62 ;
  ELSE ppp=1 ;
  hourwage_ppp = hourwage/ppp ;
  logwage_ppp = LOG(hourwage_ppp) ;
RUN ;

%IF %EVAL(&i) = 1 %THEN
  %DO ;
  DATA current ;
  SET &ccyy ;
  RUN ;
  %END ;
%ELSE
  %DO ;
  PROC APPEND BASE=current DATA=&ccyy FORCE ;
  RUN ;
  %END ;

  %LET i = %EVAL(&i+1) ;
%END ;

PROC SORT DATA=current;
  BY sex;
RUN ;

PROC SURVEYREG DATA=current;
  BY sex ;
  WEIGHT pwgt ;
  MODEL logwage_ppp =age agesq mededuc hieduc immigr livepartner youngchild oldchild
        ptime homeowner germany greece ;
RUN ;

%MEND pool ;

%LET all = us04 de04 gr04 ;
%pool

```

## Results

	Men	Women
age	0.050*** (0.0069)	0.055*** (0.0079)
agesq	-0.00050*** (0.000088)	-0.00058*** (0.000099)
mededuc	0.18*** (0.020)	0.24*** (0.021)
hieduc	0.52*** (0.022)	0.63*** (0.022)
immigr	-0.12*** (0.015)	-0.074*** (0.016)
livepartner	0.031 (0.018)	-0.0057 (0.016)
youngchild	0.077*** (0.015)	0.031 (0.019)
oldchild	0.067*** (0.013)	-0.055** (0.015)
pstime	-0.30*** (0.047)	-0.13*** (0.016)
homeowner	0.14*** (0.013)	0.088*** (0.014)
germany	0.10*** (0.012)	0.22*** (0.015)
greece	-0.37*** (0.014)	-0.21*** (0.017)
_cons	1.32*** (0.13)	0.99*** (0.15)
N	34778	33122
R-sq	0.299	0.234

Standard errors in parentheses

\*  $p < 0.05$ , \*\*  $p < 0.01$ , \*\*\*  $p < 0.001$

**Question:** How can you interpret the meaning of the coefficients for the Germany and Greece dummy variables?

- These coefficients represent the overall national level of PPP-adjusted wages, when controlling for the other variables. The small positive value for Germany reflects the fact that Germany is a slightly richer country than the United States, while the negative value for Greece reflects the fact that Greece is poorer.

**Question:** In this pooled model, which carries a higher wage penalty: being an immigrant, or working part time?

- The wage penalty for working part time is larger than that for being an immigrant. This is especially the case among men, although it should be noted that part time work is very uncommon among prime age men.